Basic Limit-state Function

Here we work with the two random variables R=capacity and S=load with the objective of determining the reliability index. The second-moment information is:

```
μR = 20;
δR = 0.1;
μS = 15;
δS = 0.2;
ρRS = 0.75;
```

The standard deviations are obtained from means and coefficients of variation:

```
σR = δR μR;
σS = δS μS;
```

Limit-state function:

```
g = R - S;
```

Gradient vector:

```
∇g = {D[g, R], D[g, S]};
MatrixForm[∇g]
```

which yields: $\begin{pmatrix} 1 \\ -1 \end{pmatrix}$

Covariance matrix:

```
Σ = {{σR^2, ρRS σR σS}, {ρRS σR σS, σS^2}};
MatrixForm[Σ]
```

which yields: $\begin{pmatrix} 4. & 4.5 \\ 4.5 & 9. \end{pmatrix}$

Visualizing the problem

When the random variables are assumed to be normally distributed then the joint probability density function is:

$$\text{jointNormalPDF} = \frac{1}{2\,\pi\,\sigma R\,\sigma S\,\sqrt{1-\rho RS^2}}\,\text{Exp}\left[-\frac{\left(\frac{r-\mu R}{\sigma R}\right)^2+\left(\frac{s-\mu S}{\sigma S}\right)^2-\frac{2\,\rho RS\,(r-\mu R)\,(s-\mu S)}{\sigma R\,\sigma S}}{2\,(1-\rho RS)^2}\right];$$

That PDF can be plotted together with the line that separates the failure domain (R<S) from the safe domain (R>S):



Reliability index by FOSM

Mean of the limit-state function:

```
μg = g /. {R → μR, S → μS}
```

which yields: 5

Standard deviation of the limit-state function:

$$\sigma g = \sqrt{\nabla g.\Sigma.\nabla g}$$

which yields: 2.

Reliability index

```
βFOSM = μg / σg
```

which yields: 2.5

With a normality assumption this corresponds to the failure probability:

```
CDF[NormalDistribution[0, 1], -βFOSM] // ScientificForm
```

which yields: 6.20967×10^{-3}

Failure probabilitiy by analytical integration

The marginal normal distributions are, under the normality assumption:

```
PDFofg = PDF[NormalDistribution[μg, σg], x];
PDFofR = PDF[NormalDistribution[μR, σR], x];
PDFofS = PDF[NormalDistribution[μS, σS], x];
```

A plot of those distributions:



Calculating the failure probability by integration:

```
∫_{-∞}^{0} PDFofg ⅆx // ScientificForm
```

which yields: 6.20967×10^{-3}

Another approach to the integral, given in the lecture notes, ONLY works when the variables are statistically independent, i.e., only when $\rho_{RS} = 0$:

```
∫_{0}^{∞} (CDF[NormalDistribution[μR, σR], x] PDFofS) ⅆx // N
```

which yields: 0.0827589

Reliability index by FORM

To transform the random variables into the standard normal space we need the Cholesky decomposition of the covariance matrix:

```
Ltilde = Transpose[CholeskyDecomposition[Σ]];
MatrixForm[Ltilde]
```

which yields: $\begin{pmatrix} 2. & 0. \\ 2.25 & 1.98431 \end{pmatrix}$

The transformation is:

```
xVector = {μR, μS} + Ltilde.{y1, y2};
MatrixForm[xVector]
```

which yields: $\begin{pmatrix} 20. + 2.\,y1 \\ 15 + 2.25\,y1 + 1.98431\,y2 \end{pmatrix}$

The transformed limit-state function is:

```
G = g /. {R → xVector[[1]], S → xVector[[2]]}
```

which yields: $5. - 0.25\,y1 - 1.98431\,y2$

The gradient in the standard normal space is:

```
∇G = {D[G, y1], D[G, y2]};
MatrixForm[∇G]
```

which yields: $\begin{pmatrix} -0.25 \\ -1.98431 \end{pmatrix}$

According to vector geometry the distance from the origin to the plane where *G*=0 is:

```
βFORM = (G /. {y1 → 0, y2 → 0}) / Norm[∇G]
```

which yields: 2.5

The corresponding failure probability is:

```
CDF[NormalDistribution[0, 1], -βFORM] // ScientificForm
```

which yields: 6.20967×10^{-3}

The alpha vector, i.e., the negative normalized gradient vector, points from the origin to the design point:

```
α = - ∇G / Norm[∇G];
MatrixForm[α]
```

which yields: $\begin{pmatrix} 0.125 \\ 0.992157 \end{pmatrix}$

That means that the design point location in the standard normal space is:

```
yStar = βFORM α;
MatrixForm[yStar]
```

which yields: $\begin{pmatrix} 0.3125 \\ 2.48039 \end{pmatrix}$

The design point coordinates in the original space of random variables is not necessarily meaningful to look at, particularly when correlation is present. It is not like in the standard normal space, where we directly view β as the distance between the origin and the design point. The concept of a distance does not exist in the original space because the random variables, in general, have different units.

```
xStar = xVector /. {y1 → yStar[[1]], y2 → yStar[[2]]}
```

which yields: {20.625, 20.625}

Plot in the original **x**-space:



Preparations for plotting in the standard normal space:

```
φ = 1/(2 π) Exp[-(y1^2 + y2^2)/2];
sol = Solve[G == 0, y2];
valueOfy2 = y2 /. sol[[1]];
```

Plot in the standard normal **y**-space:



The corresponding 3D view:



Importance vectors

For convenience we first establish the **D**-matrix, i.e., the diagonal matrix of standard deviations:

```
Dmatrix = {{σR, 0}, {0, σS}};
```

In FOSM we consider the gradient values normalized by the standard deviations:

$$\omega = -\frac{\nabla g.\text{Dmatrix}}{\text{Norm}[\nabla g.\text{Dmatrix}]} \text{ // MatrixForm}$$

which yields: $\begin{pmatrix} -0.5547 \\ 0.83205 \end{pmatrix}$

In FORM we cannot consider the $\boldsymbol{\alpha}$-vector when there is correlation present:

```
α // MatrixForm
```

which yields: $\begin{pmatrix} 0.125 \\ 0.992157 \end{pmatrix}$

Instead we must, in general, consider the $\boldsymbol{\gamma}$-vector:

$$\frac{\alpha.\text{Inverse[Ltilde].Dmatrix}}{\text{Norm}[\alpha.\text{Inverse[Ltilde].Dmatrix}]} \text{ // MatrixForm}$$

which yields: $\begin{pmatrix} -0.5547 \\ 0.83205 \end{pmatrix}$